NAG Fortran Library Routine Document

G01EEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01EEF computes the upper and lower tail probabilities and the probability density function of the beta distribution with parameters a and b.

2 Specification

```
SUBROUTINE G01EEF(X, A, B, TOL, P, Q, PDF, IFAIL)
INTEGER           IFAIL
real              X, A, B, TOL, P, Q, PDF
```

3 Description

The probability density function of the beta distribution with parameters a and b is:

$$f(B : a, b) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} B^{a-1}(1-B)^{b-1}, \quad 0 \leq B \leq 1;\ a, b > 0.$$

The lower tail probability, $\mathrm{P}(\mathrm{B} \leq \beta : a, b)$ is defined by

$$P(B \leq \beta : a, b) = \frac{\Gamma(a+b)}{\Gamma(a)\Gamma(b)} \int_0^{\beta} B^{a-1}(1-B)^{b-1} dB = I_{\beta}(a, b), \quad 0 \leq \beta \leq 1;\ a, b > 0.$$

The function $I_x(a, b)$ is also known as the incomplete beta function.

The method used is similar to that described by Majumder and Bhattacharjee (1973), and uses the following three relations for the incomplete beta function (see Abramowitz and Stegun (1972)):

$$I_x(a, b) = \frac{\Gamma(a+b)}{\Gamma(a+1)\Gamma(b)} x^a (1-x)^{b-1} + I_x(a+1, b-1) \tag{1}$$

$$I_x(a, b) = \frac{\Gamma(a+b)}{\Gamma(a+1)\Gamma(b)} x^a (1-x)^{b} + I_x(a+1, b) \tag{2}$$

$$I_x(a, b) = 1 - I_{1-x}(b, a) \tag{3}$$

If a is less than $(a+b)x$, then a and b are interchanged and $(1-x)$ replaces x, with relation (3) being used to obtain the final result.

Relation (1) is applied repeatedly until the second parameter is reduced to b', where $0 < b' \leq 1$. This produces a power series of finite length, in $x/(1-x)$, whose sum is found. If $b' = 1$, this sum equals $I_x(a, b)$, since $I_x(c, 1) = x^c/c$ for all $c > 0$.

Otherwise $(b' \neq 1)$, the integral $I_x(c, d)$ remains to be evaluated, where

$$c = a + b - b'$$
$$d = b'$$
$$0 < b' < 1.$$

Relation (2) applied repeatedly gives a convergent power series in x of infinite length.

4 References

Majumder K L and Bhattacharjee G P (1973) Algorithm AS63. The incomplete beta integral *Appl. Statist.* **22** 409–411

Abramowitz M and Stegun I A (1972) *Handbook of Mathematical Functions* (3rd Edition) Dover Publications

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Parameters

1: X – ***real*** *Input*

On entry: the value of the beta variate, β.

Constraint: $0.0 \leq \mathrm{X} \leq 1.0$.

2: A – ***real*** *Input*

On entry: the first parameter, a, of the required beta distribution.

Constraint: $0.0 < \mathrm{A} \leq 10^6$.

3: B – ***real*** *Input*

On entry: the second parameter, b, of the required beta distribution.

Constraint: $0.0 < \mathrm{B} \leq 10^6$.

4: TOL – ***real*** *Input*

On entry: the relative accuracy required by the user in the results. If G01EEF is entered with TOL greater than or equal to 1.0 or less than 10 times the ***machine precision*** (see X02AJF), then the value of $10\times$ ***machine precision*** is used instead.

5: P – ***real*** *Output*

On exit: the lower tail probability, $\mathrm{P}(\mathrm{B} \leq \beta : a, b)$.

6: Q – ***real*** *Output*

On exit: the upper tail probability, $\mathrm{P}(\mathrm{B} \geq \beta : a, b)$.

7: PDF – ***real*** *Output*

On exit: the probability density function, $f(\mathrm{B} : a, b)$.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $X < 0.0$,
or $X > 1.0$.

IFAIL = 2

On entry, $A \leq 0.0$,
or $A > 10^6$,
or $B \leq 0.0$,
or $B > 10^6$.

IFAIL = 3

The requested accuracy has not been achieved, see Section 7. Try using a larger value of TOL. The values returned for P and Q should be reasonable approximations.

IFAIL = 4

X is too far out into the tails for the probability to be evaluated exactly. The results returned are 0 and 1 as appropriate. These should be a good approximation to the required solution.

7 Accuracy

The convergence of series (2) is assumed when an upper bound on the sum of the remaining terms is less than TOL. Summation also ceases if the relative change in the sum of the series is less than ***machine precision***, in which case full accuracy cannot be guaranteed.

The accuracy is limited by the error in evaluating the logarithm of the gamma function.

8 Further Comments

The time taken by the routine will depend on the shape of the distribution. For highly skewed distributions with one of the values of a, b large and the other small, series (2) will take longer to converge than for distributions which are more symmetric.

9 Example

Values for several beta distributions are read, and the lower tail probabilities calculated and printed, until the end of data is reached.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G01EEF Example Program Text
*     Mark 15 Revised.  NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*     .. Local Scalars ..
      real             A, B, P, PDF, Q, TOL, X
      INTEGER          IFAIL
*     .. External Subroutines ..
      EXTERNAL         G01EEF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G01EEF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      WRITE (NOUT,*)
```

```
      WRITE (NOUT,*)
     +  '    X        A        B        P        Q       PDF'
      WRITE (NOUT,*)
   20 READ (NIN,*,END=40) X, A, B, TOL
      IFAIL = -1
*
      CALL G01EEF(X,A,B,TOL,P,Q,PDF,IFAIL)
*
      IF (IFAIL.EQ.0) THEN
         WRITE (NOUT,99999) X, A, B, P, Q, PDF
      ELSE
         WRITE (NOUT,99999) X, A, B, P, Q, PDF, ' NOTE: IFAIL = ', IFAIL
      END IF
      GO TO 20
   40 STOP
*
99999 FORMAT (1X,6(F7.4,2X),A,I1)
      END
```

9.2 Program Data

```
G01EEF Example Program Data
 0.25  1.0  2.0  1.9
 0.75  1.5  1.5  0.0001
 0.5   2.0  1.0  1.01
```

9.3 Program Results

```
 G01EEF Example Program Results

     X        A        B        P        Q       PDF

  0.2500   1.0000   2.0000   0.4375   0.5625   1.5000
  0.7500   1.5000   1.5000   0.8045   0.1955   1.1027
  0.5000   2.0000   1.0000   0.2500   0.7500   1.0000
```